
=====================A_Simple_Asynchronous_ADC===================================== 

1) Ten lines of spice code can model a clock-less 8bit ADC and DAC. 
2) An asynchronous ADC or AADC will track an input signal to the LSB level.
3) Such an ADC does not have a nyquist in the normal fashion. 
4) Speed limitations appear more as a truncation of the LSB bits. 
5) An easy way to build one is listed under US Patent 7,839,317 B1
 

Following the AADC with a DAC shows that input signal is being tracked down to the lSB level.

The spectrum of the AADC will consist only of the input signal and with a quantization
background noise floor. 

This type of ADC could have been invented at the same time as the 741 and 555, since it 
can use the very same NPNs and LPNPs. Imagine a 50MHz 8bit flash ADC coming out 
at about the same time and at about the same die size.  

===================MacSpiceCode====================================================
Simple_Asynchronous_ADC
*
*       VIN
*      ________________    __________    __________    __________    __________    __________    __________    
*    _|_     | TH0    _|_  | TH1    _|_  | TH0    _|_  | TH1    _|_  | TH0    _|_  | TH1    _|_  | TH0    _|_    TH1    
*   /_  \    |      _/   \ |      _/   \ |      _/   \ |      _/   \ |      _/   \ |      _/   \ |      _/   \       
*  // \  \   | |\  | \___/ | |\  | \___/ | |\  | \___/ | |\  | \___/ | |\  | \___/ | |\  | \___/ | |\  | \___/   |\  
*  \   \//   |_| \_|   |___|_| \_|   |___|_| \_|   |___|_| \_|   |___|_| \_|   |___|_| \_|   |___|_| \_|   |_____| \_
*   \___/      | / | OFF0    | / |  OFF1   | / | OFF2    | / |  OFF3   | / | OFF4    | / |  OFF5   | / | OFF6    | / |  
*    _|_       |/  |         |/  |         |/  |         |/  |         |/  |         |/  |         |/  |         |/  |
*    ///          _|_           _|_           _|_           _|_           _|_           _|_           _|_           _|_
*                |D0 |         |D1 |         |D2 |         |D3 |         |D4 |         |D5 |         |D6 |         |D7 |
*                |___|         |___|         |___|         |___|         |___|         |___|         |___|         |___|
*
*        ________
*        __|     \
*        __|      \
*        __| DAC   \__
*        __|       /
*        __|      /
*        __|_____/
*
*=========Create_Signal==================
*V_SIN#   NODE_P NODE_N DC     VALUE  SIN(   V_DC   AC_MAG FREQ   DELAY  FDamp)
VIN       VIN    0      DC     0      SIN(   .5      .5     3                  )

BTH0      D0     0      V =    u( V(VIN)  -1/2)
BOFF0     VIN    OFF0   V =    V(D0)/2
BTH1      D1     0      V =    u( V(OFF0) -1/4)
BOFF1     OFF0   OFF1   V =    V(D1)/4
BTH2      D2     0      V =    u( V(OFF1) -1/8)
BOFF2     OFF1   OFF2   V =    V(D2)/8
BTH3      D3     0      V =    u( V(OFF2) -1/16)



BOFF3     OFF2   OFF3   V =    V(D3)/16
BTH4      D4     0      V =    u( V(OFF3) -1/32)
BOFF4     OFF3   OFF4   V =    V(D4)/32
BTH5      D5     0      V =    u( V(OFF4) -1/64)
BOFF5     OFF4   OFF5   V =    V(D5)/64
BTH6      D6     0      V =    u( V(OFF5) -1/128)
BOFF6     OFF5   OFF6   V =    V(D6)/128
BTH7      D7     0      V =    u( V(OFF6) -1/256)
BOFF7     OFF6   OFF7   V =    V(D7)/245

BDAC      VOUT   0      V =    V(D0)/2+V(D1)/4+V(D2)/8+V(D3)/16+V(D4)/32+V(D5)/64+V(D6)/128+V(D7)/245

.control
*TRAN     TSTEP  TSTOP  TSTART TMAX   ?UIC?
tran      .5m    1      0      .5m   
set       pensize = 2
plot      vin  vout
plot      vin -vout
echo             "===================FFT_and_Plot====================================="
linearize 
let              FFT_BandWidth_Hz =  1k
let              FFT_resolution_Hz = 1
echo             "FFT_BandWidth_Hz=  $&FFT_BandWidth_Hz"
echo             "FFT_resolution_Hz= $&FFT_resolution_Hz"
set              specwindow =         "rectangular"
spec             $&FFT_resolution_Hz  $&FFT_BandWidth_Hz   $&FFT_resolution_Hz     v(vout)
plot             mag (vout)   loglog
echo             "===================Done============================================="
.endc
.end

